



UNIVERSIDAD DE BUENOS AIRES

FACULTAD DE INGENIERÍA

2DO CUATRIMESTRE DE 2021

[71.14 / 91.04] MODELOS Y OPTIMIZACIÓN I

CURSO 4

Trabajo Práctico 2

Empresa Agrícola

Padrón	Alumno	Email
103442	Lovera, Daniel	dlovera@fi.uba.ar
104105	Rosenblatt, Jonathan David	jrosenblatt@fi.uba.ar
103745	Martínez Quintero, Erick	egmartinez@fi.uba.ar

Índice

1. Enunciado	2
2. Análisis del problema	3
3. Objetivo	3
4. Hipótesis y supuestos	3
5. Constantes y Parámetros	4
6. Definición de variables	4
7. Modelo de programación lineal	5
8. Resolución por software	6
8.1. Modelo	6
8.1.1. .dat	6
8.1.2. .mod	8
8.2. Resultados	10
9. Informe de la solución óptima obtenida	13
10. Fe de Erratas	13

1. Enunciado

71.14 / 91.04 - Modelos y Optimización I

Turno Sábados

TRABAJO PRÁCTICO

PROBLEMA 2

La empresa agrícola Granolliers desea estudiar la política del próximo mes. Ya ha aceptado órdenes de compra de las empresas Talbott, Migueletes. y Blasco Hnos. También tiene la opción de comprar un poco de grano adicional de Granjas Solís. Los detalles de los pedidos de se presentan en la siguiente tabla.

Compañía solicitante	Talbott	Migueletes	Blasco Hnos
Cantidad [tn]	40.000 a 45.000	32.000 a 36.000	50.000 a 54.000
Humedad máxima (%)	13	15,5	15
Peso mínimo [kg/m³]	560	540	560
% máximo de daño	2	5	2
% máximo de impurezas	2	3	4
Precio de venta [U\$S/tn]	200	250	190

La compañía tiene la opción de suministrar cualquier cantidad de grano que desee, dentro del rango especificado. Claro está que deberá satisfacer los requerimientos indicados.

Granolliers mezcla los granos que le pertenecen para atender los pedidos de los clientes. Tiene almacenados 326.000 toneladas de maíz que se subdividen en 11 tipos que difieren en cuanto a (1) cantidad disponible, (2) costo por tonelada, (3) porcentaje de contenido de humedad, (4) peso por metro cúbico, (5) porcentaje de grano dañado y (6) porcentaje de impurezas. La siguiente tabla representa la información adicional acerca de las características de los distintos tipos de grano.

Tipo de maíz	Cantidad disponible [tn]	Costo [\$/tn]	Humedad (%)	Peso [kg/m³]	% daño	% impurezas
1	30.000	145	12	570	2	1,5
2	45.000	144	15	570	2	1
3	25.000	145	12	580	3	3
4	40.000	142	13	560	4	2
5	20.000	138	15	540	4	2
6	30.000	137	15	550	5	3
7	75.000	137	18	570	5	1
8	15.000	139	14	580	2	4
9	16.000	127	17	530	7	5
10	20.000	128	15	550	8	3
11	10.000	117	22	560	9	5

El grano que ofrece Granjas Solís es un cargamento de hasta 50.000 toneladas, con un promedio de 15% de humedad, 3% de daño y 2% de impurezas. La carga tiene una densidad de 570 kg por metro cúbico y el gerente de compras está convencido de que el pedido puede obtenerse a un costo de U\$S 141 por tonelada.

¿Qué es lo mejor que puede hacer Granolliers con esta información?

Nota: se debe resolver utilizando exclusivamente variables reales continuas, para poder realizar el análisis de sensibilidad en la última entrega

Segundo cuatrimestre 2021

2. Análisis del problema

Este es un problema de planificación de ventas, consiste en resolver una mezcla para distribuir todos los tipos de granos trabajando con un stock inicial sumado a la posibilidad de poder comprar al Centro Granja Solis. Debido a que la cantidad de granos almacenados en un mes es mucho mayor a la demanda y estos gastos ya fueron adquiridos, lo mejor que puede hacer Granolliers es vender los granos de la forma mas óptima para amortiguar las pérdidas que tendrá durante el mes.




Figura 1: Diagrama del proceso a resolver.

3. Objetivo

Determinar la cantidad de granos que se deben comprar a la Granja Solis y la distribución óptima de granos almacenados que se deben vender a las empresas Talbott, Migueletes y Blasco Hnos para minimizar las pérdidas de la empresa en un período mensual.

4. Hipótesis y supuestos

- Todos los granos almacenados son contabilizados en la planificación del final del mes como **costos fijos que ya fueron adquiridos**.
- Del stock inicial de granos habrán sobrantes, no se puede utilizar la totalidad de los granos por las restricciones de demanda del mes.
- Los granos de distintos tipos destinados a cada empresa son homogéneos, no hay distinción física entre los mismos.
- Los granos se distribuyen uniformemente, una muestra del total de granos conservara los mismos valores de humedad, impureza, daño, densidad y costo.
- La cantidad total de granos destinados a una empresa es indistinguible al momento de la entrega, se contabilizan como un todo.

- No hay pérdidas de granos al momento de realizar las mezclas.
- La estimación del gerente de compras en el costo de granos de Granja Solis es acertada.
- Se pueden mezclar cantidades fraccionarias de granos.
- La empresa no se hace cargo de costos adicionales o cualquier tipo de mano de obra.
- No hay ningún tipo de fenómeno monetario que pueda afectar la variación de precios.
- Los granos almacenados no sufren variaciones en sus propiedades a lo largo del tiempo, se mantendrán al menos hasta fin de mes.

5. Constantes y Parámetros

Los subíndices i y j son iguales a nombres en formato texto, sin embargo, para cada nombre habrá un mapeo interno a números comenzando con 1. Por ejemplo: Talbott = 1, Migueletes = 2, Blasco Hnos = 3.

Sea i = Talbott, Migueletes, Blasco Hnos.

Sea j = Tipo1, Tipo2, Tipo3, ..., Tipo11, Granja Solis.

1. **pi**: Precio de venta de granos destinados a j por tonelada [U\$S / tn]
2. **cj**: Costos de granos de tipo j en un mes [U\$S / tn]
3. **gj**: Granos de tipo j disponibles en un mes [tn / mes]
4. **g**: Granos totales disponibles en un mes [tn / mes]
5. **hj**: Humedad del grano tipo j [%]
6. **vj**: Densidad del grano de tipo j [kg / m³]
7. **dj**: Daño del grano de tipo j [%]
8. **ij**: Impurezas del grano de tipo j [%]

6. Definición de variables

Dirigirse a **Constantes y Parámetros** para información sobre subíndices. Adicionalmente todas las variables utilizadas son continuas.

1. **Ci**: Cantidad de granos destinados a i en un mes [tn/mes]
2. **Hi**: Toneladas de granos húmedos destinados a i en un mes, si se considera que 1 ton = 1 m³ se puede obtener el volumen de humedad en los granos destinados a i [ton / mes]
3. **Vi**: Volumen de granos destinados a i en un mes [m³/mes]
4. **Di**: Daño de granos destinados a i en un mes [tn/mes]
5. **Ii**: Impurezas de granos destinados a i en un mes [tn/mes]
6. **GPji**: Granos parciales de tipo j destinados a i en un mes [tn/mes]
7. **GTj**: Granos totales de tipo j consumidos en un mes [tn/mes]

7. Modelo de programación lineal

Restricciones de las órdenes de compra:

1. Demanda máxima de granos [tn / mes]:

$$\begin{aligned} C1 &\leq 45000 \\ C2 &\leq 36000 \\ C3 &\leq 54000 \end{aligned}$$
2. Demanda mínima de granos [tn / mes]:

$$\begin{aligned} C1 &\geq 40000 \\ C2 &\geq 32000 \\ C3 &\geq 50000 \end{aligned}$$
3. Humedad máxima de granos: [vol humedad / mes]

$$\begin{aligned} H1 &\leq 0.13 * C1 \\ H2 &\leq 0.155 * C2 \\ H3 &\leq 0.15 * C3 \end{aligned}$$
4. Volumen mínimo de granos: [m³ / mes]]

$$\begin{aligned} V1 &\geq C1 * (1000 / 560) \\ V2 &\geq C2 * (1000 / 540) \\ V3 &\geq C3 * (1000 / 560) \end{aligned}$$
5. Daño máximo de granos: [tn / mes]

$$\begin{aligned} D1 &\leq 0.02 * C1 \\ D2 &\leq 0.05 * C2 \\ D3 &\leq 0.02 * C3 \end{aligned}$$
6. Impureza máxima de granos: [tn / mes]

$$\begin{aligned} I1 &\leq 0.02 * C1 \\ I2 &\leq 0.03 * C2 \\ I3 &\leq 0.04 * C3 \end{aligned}$$

Vínculos (Entradas y Salidas):

1. Vínculos entre granos de tipo j y granos de tipo j destinados a i:

$$\begin{aligned} GT1 &= GP11 + GP12 + GP13 \\ GT2 &= GP21 + GP22 + GP23 \\ GT3 &= GP31 + GP32 + GP33 \\ &\vdots \\ GT11 &= GP11\ 1 + GP11\ 2 + GP11\ 3 \\ GT12 &= GP12\ 1 + GP12\ 2 + GP12\ 3 \end{aligned}$$
2. Vínculos entre granos de tipo j destinados a i y cantidad de granos destinados a i :

$$\begin{aligned} GP11 + GP21 + GP31 + \dots + GP11\ 1 + GP12\ 1 &= C1 \\ GP12 + GP22 + GP32 + \dots + GP11\ 2 + GP12\ 2 &= C2 \\ GP13 + GP23 + GP33 + \dots + GP11\ 3 + GP12\ 3 &= C3 \end{aligned}$$

Relaciones:

1. Relación granos de tipo j y granos de tipo j destinados a i:

$$g1 \geq GP11 + GP12 + GP13$$

```

g2 >= GP21 + GP22 + GP23
g3 >= GP31 + GP32 + GP33
.
.
.
g11 >= GP11 1 + GP11 2 + GP11 3
g12 >= GP12 1 + GP12 2 + GP12 3

```

2. Relación de humedad:

$$H_i = \sum_{j=1}^{12} [(h_j/100) * GP_{ji}]$$

3. Relación de volumen:

$$V_i = \sum_{j=1}^{12} [GP_{ji} * (1000 / v_j)]$$

4. Relación de daños:

$$D_i = \sum_{j=1}^{12} [(d_j/100) * GP_{ji}]$$

5. Relación de impurezas:

$$I_i = \sum_{j=1}^{12} [(i_j/100) * GP_{ji}]$$

Funcional:

```

Ventas = p1 * C1 + p2 * C2 + p3 * C3
Costos FIJOS de insumos en stock = c1 * g1 + c2 * g2 + ... + c11 * g11
Costos de Granos Solis = c12 * GT12

```

$$\text{MIN } Z = \text{Costos} - \text{Ventas}$$

$$\text{MIN } Z = (\text{Costos de FIJOS de insumos en stock} + \text{Costos de Granos Solis}) - \text{Ventas}$$

8. Resolución por software

8.1. Modelo

8.1.1. .dat

```

data;

set granos := G1 G2 G3 G4 G5 G6 G7 G8 G9 G10 G11 G12;
set destino := D1 D2 D3;

param precio_granos :=
    G1 145
    G2 144
    G3 145
    G4 142
    G5 138
    G6 137
    G7 137
    G8 139
    G9 127
    G10 128
    G11 117
    G12 141;

```

```
param disponibilidad_granos := G1 30000
                               G2 45000
                               G3 25000
                               G4 40000
                               G5 20000
                               G6 30000
                               G7 75000
                               G8 15000
                               G9 16000
                               G10 20000
                               G11 10000
                               G12 50000;
```

```
param humedad_granos := G1 12
                        G2 15
                        G3 12
                        G4 13
                        G5 15
                        G6 15
                        G7 18
                        G8 14
                        G9 17
                        G10 15
                        G11 22
                        G12 14;
```

```
param densidad_granos := G1 570
                          G2 570
                          G3 580
                          G4 560
                          G5 540
                          G6 550
                          G7 570
                          G8 580
                          G9 530
                          G10 550
                          G11 560
                          G12 570;
```

```
param dano_granos := G1 2
                     G2 2
                     G3 3
                     G4 4
                     G5 4
                     G6 4
                     G7 5
                     G8 2
                     G9 7
                     G10 8
                     G11 9
                     G12 3;
```

```
param impureza_granos := G1 1.5
                          G2 1
```



```

                                G3 3
                                G4 2
                                G5 2
                                G6 3
                                G7 1
                                G8 4
                                G9 5
                                G10 3
                                G11 5
                                G12 2;

param demanda_minima := D1 40000
                        D2 32000
                        D3 50000;

param demanda_maxima := D1 45000
                        D2 36000
                        D3 54000;

param humedad_maxima := D1 13
                        D2 15.5
                        D3 15;

param densidad_minima := D1 560
                        D2 540
                        D3 560;

param dano_maximo := D1 2
                    D2 5
                    D3 2;

param impureza_maxima := D1 2
                        D2 3
                        D3 4;

param precio_venta := D1 200
                       D2 250
                       D3 190;

end;
```

8.1.2. .mod

```
/* TP2 Empresa Agricola
 *
 * Compilacion del modelo:
 *      glpsol -m tp2.mod -d tp2.dat -o tp2.sol
 *
 * Observacion: Los archivos .mod y .dat deberan estar en
 * el mismo directorio, esto generara un archivo tp2.sol
 * con los resultados de la corrida.
 */

set granos;
```

```

set destino;

param precio_granos {i in granos};
param disponibilidad_granos {i in granos};
param humedad_granos {i in granos};
param densidad_granos {i in granos};
param dano_granos {i in granos};
param impureza_granos {i in granos};

param demanda_minima {i in destino};
param demanda_maxima {i in destino};
param humedad_maxima {i in destino};
param densidad_minima {i in destino};
param dano_maximo {i in destino};
param impureza_maxima {i in destino};
param precio_venta {i in destino};

var C {i in destino} >= 0;
var H {i in destino} >= 0;
var V {i in destino} >= 0;
var D {i in destino} >= 0;
var I {i in destino} >= 0;
var GP {i in granos, j in destino} >= 0;
var GT {i in granos} >= 0;

/* Demandas maximas */
s.t. dem_max {i in destino}: C[i] <= demanda_maxima[i];

/* Demandas minimas */
s.t. dem_min {i in destino}: C[i] >= demanda_minima[i];

/* Humedad maxima */
s.t. hum_max {i in destino}: H[i] <= C[i] *
                                (humedad_maxima[i] / 100);

/* Volumen minimo de granos */
s.t. vol_min {i in destino}: V[i] >= C[i] /
                                (1000 / densidad_minima[i]);

/* Danio maximo de granos: */
s.t. dan_max {i in destino}: D[i] <= C[i] *
                                (dano_maximo[i] / 100);

/* Impureza maxima de granos */
s.t. imp_max {i in destino}: I[i] <= C[i] *
                                (impureza_maxima[i] / 100);

/* Relacion de granos totales y parciales */
s.t. ES_gnos {i in granos}: GT[i] = sum {j in destino} GP[i,j];

/* Relacion granos de tipo j y granos de tipo j destinados a i */
s.t. dp_gnos {i in granos}: sum {j in destino} GP[i,j] <=
                                disponibilidad_granos[i];

```

```

/* Relacion de granos */
s.t. ES_dest{j in destino}: C[j] = sum{i in granos} GP[i, j];

/*Relacion de humedad*/
s.t. rel_hum{j in destino}: H[j] = sum{i in granos} GP[i, j] *
                                (humedad_granos[i] / 100);

/*Relacion de densidad*/
s.t. rel_den{j in destino}: V[j] = sum{i in granos} GP[i, j] /
                                (1000 / densidad_granos[i]);

/*Relacion de danos*/
s.t. rel_dan{j in destino}: D[j] = sum{i in granos} GP[i, j] *
                                (dano_granos[i] / 100);

/*Relacion de impureza*/
s.t. rel_imp {j in destino}: I[j] = sum{i in granos} GP[i, j] *
                                (impureza_granos[i] / 100);

minimize Z: (sum{i in granos: i<>'G12'} precio_granos[i] *
             disponibilidad_granos[i])
            + (precio_granos['G12'] * GT['G12'])
            - (sum{j in destino} precio_venta[j] * C[j]);

```

8.2. Resultados

Se eliminaron las columnas correspondientes a lower bound y upper bound del resultado en el modelo, para que pueda ser incluido en su totalidad los puntos mas importantes de la solución.

```

Problem:      tp2
Rows:         58
Columns:      63
Non-zeros:    313
Status:       OPTIMAL
Objective:    Z = 18627000 (MINimum)

```

No.	Row name	St	Activity	Marginal
1	dem_max[D1]	B	40000	
2	dem_max[D2]	NU	36000	-250
3	dem_max[D3]	B	50000	
4	dem_min[D1]	B	40000	
5	dem_min[D2]	B	36000	
6	dem_min[D3]	NL	50000	10
7	hum_max[D1]	B	-100	
8	hum_max[D2]	NU	0	< eps
9	hum_max[D3]	B	-150	
10	vol_min[D1]	B	400	
11	vol_min[D2]	B	1230	
12	vol_min[D3]	B	650	
13	dan_max[D1]	NU	0	-20000
14	dan_max[D2]	B	-300	
15	dan_max[D3]	NU	0	-20000
16	imp_max[D1]	B	-250	
17	imp_max[D2]	B	-420	

18	imp_max[D3]	B	-1050	
19	ES_gnos[G1]	NS	0	< eps
20	ES_gnos[G2]	NS	0	< eps
21	ES_gnos[G3]	NS	0	< eps
22	ES_gnos[G4]	NS	0	< eps
23	ES_gnos[G5]	NS	0	< eps
24	ES_gnos[G6]	NS	0	< eps
25	ES_gnos[G7]	NS	0	< eps
26	ES_gnos[G8]	NS	0	< eps
27	ES_gnos[G9]	NS	0	< eps
28	ES_gnos[G10]	NS	0	< eps
29	ES_gnos[G11]	NS	0	< eps
30	ES_gnos[G12]	NS	0	141
31	dp_gnos[G1]	NU	30000	-200
32	dp_gnos[G2]	NU	45000	-200
33	dp_gnos[G3]	B	15000	
34	dp_gnos[G4]	B	0	
35	dp_gnos[G5]	B	0	
36	dp_gnos[G6]	B	0	
37	dp_gnos[G7]	B	21000	
38	dp_gnos[G8]	NU	15000	-200
39	dp_gnos[G9]	B	0	
40	dp_gnos[G10]	B	0	
41	dp_gnos[G11]	B	0	
42	dp_gnos[G12]	B	0	
43	ES_dest[D1]	NS	0	-600
44	ES_dest[D2]	NS	0	< eps
45	ES_dest[D3]	NS	0	-600
46	rel_hum[D1]	NS	0	< eps
47	rel_hum[D2]	NS	0	< eps
48	rel_hum[D3]	NS	0	< eps
49	rel_den[D1]	NS	0	< eps
50	rel_den[D2]	NS	0	< eps
51	rel_den[D3]	NS	0	< eps
52	rel_dan[D1]	NS	0	20000
53	rel_dan[D2]	NS	0	< eps
54	rel_dan[D3]	NS	0	20000
55	rel_imp[D1]	NS	0	< eps
56	rel_imp[D2]	NS	0	< eps
57	rel_imp[D3]	NS	0	< eps
58	Z	B	-2.65e+07	

No.	Column name	St	Activity	Marginal
1	C[D1]	B	40000	
2	C[D2]	B	36000	
3	C[D3]	B	50000	
4	H[D1]	B	5100	
5	H[D2]	B	5580	
6	H[D3]	B	7350	
7	V[D1]	B	22800	
8	V[D2]	B	20670	
9	V[D3]	B	28650	
10	D[D1]	B	800	

11	D[D2]	B	1500	
12	D[D3]	B	1000	
13	I[D1]	B	550	
14	I[D2]	B	660	
15	I[D3]	B	950	
16	GP[G1,D1]	B	30000	
17	GP[G1,D2]	NL	0	200
18	GP[G1,D3]	NL	0	< eps
19	GP[G2,D1]	B	10000	
20	GP[G2,D2]	NL	0	200
21	GP[G2,D3]	B	35000	
22	GP[G3,D1]	B	0	
23	GP[G3,D2]	B	15000	
24	GP[G3,D3]	B	0	
25	GP[G4,D1]	NL	0	200
26	GP[G4,D2]	NL	0	< eps
27	GP[G4,D3]	NL	0	200
28	GP[G5,D1]	NL	0	200
29	GP[G5,D2]	NL	0	< eps
30	GP[G5,D3]	NL	0	200
31	GP[G6,D1]	NL	0	200
32	GP[G6,D2]	NL	0	< eps
33	GP[G6,D3]	NL	0	200
34	GP[G7,D1]	NL	0	400
35	GP[G7,D2]	B	21000	
36	GP[G7,D3]	NL	0	400
37	GP[G8,D1]	NL	0	< eps
38	GP[G8,D2]	NL	0	200
39	GP[G8,D3]	B	15000	
40	GP[G9,D1]	NL	0	800
41	GP[G9,D2]	NL	0	< eps
42	GP[G9,D3]	NL	0	800
43	GP[G10,D1]	NL	0	1000
44	GP[G10,D2]	NL	0	< eps
45	GP[G10,D3]	NL	0	1000
46	GP[G11,D1]	NL	0	1200
47	GP[G11,D2]	NL	0	< eps
48	GP[G11,D3]	NL	0	1200
49	GP[G12,D1]	NL	0	141
50	GP[G12,D2]	NL	0	141
51	GP[G12,D3]	NL	0	141
52	GT[G1]	B	30000	
53	GT[G2]	B	45000	
54	GT[G3]	B	15000	
55	GT[G4]	B	0	
56	GT[G5]	B	0	
57	GT[G6]	B	0	
58	GT[G7]	B	21000	
59	GT[G8]	B	15000	
60	GT[G9]	B	0	
61	GT[G10]	B	0	
62	GT[G11]	B	0	
63	GT[G12]	B	0	

Karush–Kuhn–Tucker optimality conditions :

KKT.PE: max.abs.err = 1.46e−11 on row 45
 max.rel.err = 2.27e−16 on row 15
 High quality

KKT.PB: max.abs.err = 0.00e+00 on row 0
 max.rel.err = 0.00e+00 on row 0
 High quality

KKT.DE: max.abs.err = 3.64e−12 on column 10
 max.rel.err = 6.32e−14 on column 23
 High quality

KKT.DB: max.abs.err = 5.68e−14 on column 37
 max.rel.err = 5.68e−14 on column 37
 High quality

End of output

9. Informe de la solución óptima obtenida

Considerando el objetivo planteado, lo mejor que puede hacer la empresa Granollers es evitar comprar granos a la Granja Solis y distribuir los granos que ya tienen almacenados de la siguiente manera:

- Destinar 30000 ton de grano Tipo1 a la empresa Talbott
- Destinar 10000 ton de grano Tipo2 a la empresa Talbott
- Destinar 15000 ton de grano Tipo3 a la empresa Migueletes
- Destinar 21000 ton de grano Tipo7 a la empresa Migueletes
- Destinar 35000 ton de grano Tipo2 a Blasco Hnos
- Destinar 15000 ton de grano Tipo8 a la empresa Blasco Hnos

Con esta distribución la empresa podrá cumplir con las órdenes de compra aceptadas dentro de las restricciones pactadas y minimizará las pérdidas en el mes, tomando en cuenta que adquirió una gran deuda por comprar 326000 ton en granos al inicio de la planificación.

10. Fe de Erratas

- En el gráfico de la **Figura 1** las entradas g1, g2, g3 corresponden a variables desconocidas en principio, por lo tanto la denominación correcta es GP1, GP2,...,GP11.
- Las variables Gji y Gj se encuentran mal expresadas con respecto a la notación explicada en **definición de variables**, lo correcto es GPji y GTj.